65809
InnoSetupのスクリプトを作成しています。  
スタートメニューにヘルプファイルやマニュアルのような実行ファイルと別のファイルを登録したいのですが、記述方法がわかりません。  
サンプルのスクリプトファイルを参照しながら作成していますが、理解が追い付きません。  
以下のスクリーンショットの赤丸で囲ったような追加をしたいと考えています。  
ご存じの方はご教示いただけないでしょうか。

[![スタートメニューのサンプル](https://i.stack.imgur.com/6dPep.jpg)](https://i.stack.imgur.com/6dPep.jpg)
CC BY-SA 4.0
2020-04-20T08:19:48.653
65781
2020-04-21T11:41:23.927
2020-04-20T08:29:16.683
3060
32891
question
windows-10 inno-setup
Inno Setup でスタートメニューに複数のファイルを表示させる方法
コメント欄にて提示していただいたURLを参照して解決いたしました。

[Inno Setup Create more icons on Start menu (Win10) -
StackOverflow](https://stackoverflow.com/q/49496494)

`#define` による宣言と、`[Icons]` での `{group}` を記述すれば大丈夫でした。  
App_id部分を省略したスクリプトのコードは以下の通りです。

```

    #define MyAppName "MyApp"
    #define MyAppVersion "1.0"
    #define MyAppPublisher "MyCompany"
    #define MyAppExeName "MyApp.exe"
    #define AppSrcPath "dist\MyApp"
    #define MyAppSetupName "MyApp_setup"
    #define MyAppManualName "Manual"
    #define MyAppManualPDFName "manual\manual.pdf"
    
    [Setup]
    -----
    AppName={#MyAppName}
    AppVersion={#MyAppVersion}
    AppPublisher={#MyAppPublisher}
    DefaultDirName={autopf}\{#MyAppName}
    DefaultGroupName={#MyAppName}
    DisableProgramGroupPage=yes
    OutputBaseFilename={#MyAppSetupName}
    Compression=lzma
    SolidCompression=yes
    WizardStyle=modern
    SetupIconFile=.\logo\MyApp.ico
    UninstallDisplayIcon=.\logo\MyApp.ico
    
    [Languages]
    Name: "japanese"; MessagesFile: "compiler:Languages\Japanese.isl"
    
    [Tasks]
    Name: "desktopicon"; Description: "{cm:CreateDesktopIcon}"; GroupDescription: "{cm:AdditionalIcons}"; Flags: unchecked
    
    [Files]
    Source: "{#SourcePath}{#AppSrcPath}\MyApp.exe"; DestDir: "{app}"; Flags: ignoreversion
    Source: "{#SourcePath}{#AppSrcPath}\*"; DestDir: "{app}"; Flags: ignoreversion recursesubdirs createallsubdirs
    
    [Icons]
    Name: "{group}\{#MyAppName}"; Filename: "{app}\{#MyAppExeName}"
    Name: "{group}\{#MyAppManualName}"; Filename: "{app}\{#MyAppManualPDFName}"
    Name: "{autodesktop}\{#MyAppName}"; Filename: "{app}\{#MyAppExeName}"; Tasks: desktopicon
    
    [Run]
    Filename: "{app}\{#MyAppExeName}"; Description: "{cm:LaunchProgram,{#StringChange(MyAppName, '&', '&&')}}"; Flags: nowait postinstall skipifsilent
    
```
CC BY-SA 4.0
2020-04-21T05:23:31.217
65809
2020-04-21T11:41:23.927
2020-04-21T11:41:23.927
3060
32891
65781
answer